

Program for Item-by-Item Sequential Sampling by Attributes

F.A. BURNEY and A.K.M. ABDUL HAQ
*Industrial Engineering Department,
Faculty of Engineering, King Abdulaziz University,
Jeddah, Saudi Arabia.*

ABSTRACT. This computer program, which is written in BASIC Language for an IBM-PC or compatible computer, generates solutions to sequential sampling problems. Given the acceptable quality level, the producer's risk, the lot tolerance fraction defective and the consumer's risk, the program produces all or one of the following :

- i) Graphical representation of the plan (provided the computer has the color graphics capability).
- ii) Table of acceptance and rejection numbers.
- iii) Operational characteristic curve.
- iv) Average sample number curve.

Furthermore, the program has a unique feature of calculating the values of probability of acceptance and average sample number for any value of proportion defective between acceptable quality level and lot tolerance fraction defective.

KEY WORDS: Item-by-Item Sequential Sampling Plan, Acceptable Quality Level, Producer's Risk, Lot Tolerance Fraction Defective, Consumer's Risk, Operating Characteristic Curve, Average Sample Number Curve, Sequential Probability Ratio.

1. Introduction

Acceptance sampling plans have become an important part of statistical quality control. These plans include the whole spectrum of attribute and variable types of plans. In the attribute type there has been a successive development from single to double to multiple to sequential sampling plans, with corresponding reduction in the cost of sampling^[1].

In a sequential sampling plan a sequence of samples is taken from the lot and the number of samples is determined by the result of sampling process. Although, theoretically, this procedure can continue indefinitely until the whole lot is inspected, it is generally truncated after the number inspected is equal to three times the number that would have been inspected in a single sampling plan with the same operating characteristic curve. When the sample size is one, the plan is called item-by-item sequential sampling plan^[2].

Since calculation of plan variables and their subsequent plotting requires considerable computation work, a number of computer programs have been developed for them. However, perhaps because of the relative ease, more effort has been directed towards the single sampling plan^[3-5]. For double sampling plan two separate programs have been evolved^[5-6]. As for the item-by-item sequential sampling plan a simple program showing only the acceptance and rejection numbers has been presented in Collins *et al.*^[5]. It does not plot the operating characteristic and average sampling number curves. Nor does it incorporate the calculation of probability of acceptance and average sample number for various values of proportion defective.

This paper is an attempt to make up these deficiencies for an item-by-item sequential sampling plan. It presents an interactive computer program written in BASIC language for a IBM-PC or compatible computer. The program generates a complete solution to an item-by-item sequential sampling problem.

The paper describes briefly the working of an item-by-item sequential sampling plan along with a solved case study. It then briefly describes the program. An example of running the program and the program listing are also included.

2. Description of Item-by-Item Sequential Sampling Plan

This plan is based on the concept of Sequential Probability Ratio suggested by Wald^[7]. It is the ratio of the probability of getting the cumulated sample result if the material was of barely tolerable quality to the probability of getting this result if the material was of acceptable quality. Wald showed that for operational purposes it amounts to drawing an acceptance and a rejection line on a graph with abscissa as the total number of items inspected and ordinate as the number of defectives observed (Fig. 1).

Samples continue to be drawn as long as the plot stays within these lines. Hitting or crossing a line results in a decision – acceptance for the lower line and rejection for the higher line. Furthermore, a table corresponding to this graph can also be developed. As with other sampling plans by attributes, there is an operating characteristic curve (probability of acceptance vs proportion defective) and average sample number curve (average sample number vs proportion defective).

Given a set of acceptable quality level (p_1), the producer's risk (α), the lot toler-

ance fraction defective (p_2), and the consumer's risk (β), the equations of acceptance and rejection lines are given as follows




FIG. 1. Graphical representation of Wald's item-by-item sequential sampling plan.

Acceptance line $Y = -h_1 + S$ (1)

Rejection line $Y = h_2 + S$ (2)

where

Y = cumulative number of defective items found.

n = cumulative number of items inspected.

$$h_1 = \log \left(\frac{1 - \alpha}{\beta} \right) / \log \left[\frac{p_2(1 - p_1)}{p_1(1 - p_2)} \right] \quad (3)$$

$$h_2 = \log \left(\frac{1 - \beta}{\alpha} \right) / \log \left[\frac{p_2(1 - p_1)}{p_1(1 - p_2)} \right] \quad (4)$$

$$S = \log \left[\frac{1 - p_1}{1 - p_2} \right] / \log \left[\frac{p_2(1 - p_1)}{p_1(1 - p_2)} \right] \quad (5)$$

For the operating characteristic (OC) curve three points are obtained as follows

Probability of acceptance	Proportion defective
$1 - \alpha$	p_1
$\frac{h_2}{h_1 + h_2}$	S
β	p_2

For the average sample number (ASN) curve four points are obtained as follows

Average sample number	Proportion defective
h_1 / S	0
$\frac{(1 - \alpha) h_1 - \alpha h_2}{S - p_1}$	p_1
$\frac{h_1 h_2}{S(1 - S)}$	S
$\frac{(1 - \beta) h_2 - \beta h_1}{p_2 - S}$	p_2

Other points for these two curves can be obtained by varying the value of θ in the following equations.

$$\text{Proportion defective } p = \frac{1 - \left(\frac{1 - p_2}{1 - p_1}\right)^\theta}{\left(\frac{p_2}{p_1}\right)^\theta - \left(\frac{1 - p_2}{1 - p_1}\right)^\theta} \quad (6)$$

$$\text{Probability of acceptance, } p_a = \frac{\left(\frac{1 - \beta}{\alpha}\right)^\theta - 1}{\left(\frac{1 - \beta}{\alpha}\right)^\theta - \left(\frac{\beta}{1 - \alpha}\right)^\theta} \quad (7)$$

$$\text{Average sample number, } ASN = \frac{p_a \log\left(\frac{\beta}{1 - \alpha}\right) + (1 - p_a) \log\left(\frac{1 - \beta}{\alpha}\right)}{p \log\left(\frac{p_2}{p_1}\right) + (1 - p) \log\left(\frac{1 - p_2}{1 - p_1}\right)} \quad (8)$$

For $\theta = 1$, $p = p_1$, for $\theta = 0$, $p = S$, and for $\theta = -1$, $p = p_2$. Hence $\theta = \pm 0.5$ gives the good intermediate values which have been used in this program both for OC curve and ASN curve. For the latter, an additional point has been calculated corresponding to $p = p_1 + p_2$.

One of the unique features of this paper is that, with the help of the above equations and by trial and error method of choosing the value of θ , it calculates the

- a) probability of acceptance, and
- b) average sample number,

for any proportion defective between acceptable quality level and lot tolerance fraction defective within ± 0.001 . This approach did not exist in literature before.

3. Case Study

In order to illustrate the above description, results of a case study where $p_1 = 0.015$, $p_2 = 0.07$, $\alpha = 0.05$, and $\beta = 0.10$ are given.

$$h_1 = 1.409, h_2 = 1.809, S = 0.036$$

$$\text{Acceptance line, } Y = -1.409 + 0.036 n$$

$$\text{Rejection line, } Y = 1.809 + 0.036 n$$

Proportion defective	Probability of acceptance	Average sample number
0	–	39.13
0.015	0.950	60.00
0.024	0.827	70.40
0.036	0.562	73.44
0.051	0.268	60.90
0.070	0.100	44.00
0.085	–	33.40

4. Program Operation

After loading the program, as the run key is pressed an option is available asking if an introduction to the program is required or not. The program then asks for values of p_1 , α , p_2 and β . It then prints the possible options available which are as follows

- Print a graph (to be used only if the computer is equipped with color graphics).
- Print a table of acceptance and rejection numbers (up to 80 items inspected).
- Plot five points of operating characteristic curve.
- Plot seven points of average sample number curve.
- Calculate values of probability of acceptance and average sample number for any value of proportion defective between acceptable quality level and lot tolerance fraction defective.
- Start a new problem.
- Exit from the program.

Every time that an option is chosen and the necessary output obtained, all of the above seven options again become available.

Example

The following is a complete example of running this program.

A listing of the program is attached at the end.

```

*****
*      PROGRAM FOR SEQUENTIAL SAMPLING PLAN      *
*              DEVELOPED BY                      *
*              DR. F.A.BURNEY                   *
*              ASSOCIATE PROFESSOR             *
*              AND                              *
*              DR. A.K.M.ABDUL HAQ            *
*              ASSISTANT PROFESSOR             *
*              INDUSTRIAL ENGINEERING DEPARTMENT *
*              COLLEGE OF ENGINEERING         *
*              KING ABDULAZIZ UNIVERSITY      *
*              JEDDAH. SAUDI ARABIA          *
*****

```

NEED ANY INTRODUCTION (Y OR N) ? Y

THIS PROGRAM IS ABOUT SEQUENTIAL SAMPLING PLAN AND IS BASED ON WALD'S SEQUENTIAL PROBABILITY RATIO. GIVEN THE VALUES OF THE PRODUCERS RISK(ALPHA), THE CONSUMERS RISK(BETA), THE ACCEPTABLE QUALITY LEVEL,AND THE LOT TOLERANCE FRACTION DEFECTIVE, THE PLAN WILL BE FULLY DEFINED BY THE EQUATIONS OF THE FOLLOWING TWO LINES.

LINE OF ACCEPTANCE

LINE OF REJECTION

INPUT REQUIREMENTS

- 1.- ACCEPTABLE QUALITY LEVEL (AQL)
- 2.- PRODUCER'S RISK (ALPHA VALUE)
- 3.- LOT TOLERANCE FRACTION DEFECTIVE (LTFD)
- 4.- CONSUMER'S RISK (BETA VALUE)

ENTER ACCEPTABLE QUALITY LEVEL (AQL)....? .015
 ENTER PRODUCER'S RISK (ALPHA VALUE).....? .05
 ENTER LOT TOLERANCE FRACTION DEFECTIVE (LTFD)...? .07
 ENTER CONSUMER'S RISK (BETA VALUE).....? .1

AVAILABLE OPTIONS

- 1) TO PRINT A GRAPHICAL REPRESENTATION OF THE PLAN PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS.
- 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE AND REJECTION NUMBERS
- 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE

- 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE
- 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA)
AND ASN FOR A PROPORTION DEFECTIVE (P)
BETWEEN AQL AND LTFD
- 6) TO START A NEW PROBLEM
- 7) TO EXIT FROM THE PROGRAM

ENTER YOUR SELECTION OF THE ABOVE OPTIONS ? 1

IN THE DIAGRAM BELOW THERE ARE THREE REGIONS NAMELY:
ACCEPTANCE REGION BELOW THE BOTTOM LINE, REJECTION REGION
ABOVE THE TOP LINE, AND CONTINUE-SAMPLING REGION IN
BETWEEN THE TWO LINES. FURTHERMORE, THE X-AXIS INDICATES
THE NUMBER OF ITEMS SAMPLED AND THE Y-AXIS THE NUMBER
OF DEFECTIVES.

PLEASE ALSO NOTE THAT THIS DIAGRAM CAN BE PRINTED BY
USING THE CAPSHIFT AND PRISC KEYS TOGETHER.

PRESS ANY KEY TO CONTINUE

0
1
2
3
4
0 10 20 30 40 50 60 REJ. REGION
ACC. REGION

PRESS ANY KEY TO CONTINUE

AVAILABLE OPTIONS

- 1) TO PRINT A GRAPHICAL REPRESENTATION OF THE PLAN
PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF
YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS.
 - 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE
AND REJECTION NUMBERS
 - 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE
 - 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE
-

- 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA)
AND ASN FOR A PROPORTION DEFECTIVE (P)
BETWEEN AQL AND LTFD
- 6) TO START A NEW PROBLEM
- 7) TO EXIT FROM THE PROGRAM

ENTER YOUR SELECTION OF THE ABOVE OPTIONS ? ?

 TABLE OF ACCEPTANCE & REJECTION NOS. FOR SEQUENTIAL SAMPLING
 PLAN WITH AQL= .015 .LTFD= .07 .ALPHA= .05 .BETA= .1 AND
 UPTO A SAMPLE SIZE OF 80. IN THE TABLE BELOW N IS THE SAMPLE
 SIZE, A THE ACCEPTANCE NUMBER, AND R THE REJECTION NUMBER.

: N	A	R	: N	A	R	: N	A	R	: N	A	R
: 1	*	2	: 2	*	2	: 3	*	2	: 4	*	2
: 5	*	2	: 6	*	3	: 7	*	3	: 8	*	3
: 9	*	3	: 10	*	3	: 11	*	3	: 12	*	3
: 13	*	3	: 14	*	3	: 15	*	3	: 16	*	3
: 17	*	3	: 18	*	3	: 19	*	3	: 20	*	3

: 21	*	3	: 22	*	3	: 23	*	3	: 24	*	3
: 25	*	3	: 26	*	3	: 27	*	3	: 28	*	3
: 29	*	3	: 30	*	3	: 31	*	3	: 32	*	3
: 33	*	3	: 34	*	4	: 35	*	4	: 36	*	4
: 37	*	4	: 38	*	4	: 39	*	4	: 40	0	4

: 41	0	4	: 42	0	4	: 43	0	4	: 44	0	4
: 45	0	4	: 46	0	4	: 47	0	4	: 48	0	4
: 49	0	4	: 50	0	4	: 51	0	4	: 52	0	4
: 53	0	4	: 54	0	4	: 55	0	4	: 56	0	4
: 57	0	4	: 58	0	4	: 59	0	4	: 60	0	4

: 61	0	5	: 62	0	5	: 63	0	5	: 64	0	5
: 65	0	5	: 66	0	5	: 67	1	5	: 68	1	5
: 69	1	5	: 70	1	5	: 71	1	5	: 72	1	5
: 73	1	5	: 74	1	5	: 75	1	5	: 76	1	5
: 77	1	5	: 78	1	5	: 79	1	5	: 80	1	5

 PRFSS ANY KEY TO CONTINUE

AVAILABLE OPTIONS

- 1) TO PRINT A GRAPHICAL REPRESENTATION OF THE PLAN
PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF
YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS.

- 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE AND REJECTION NUMBERS
- 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE
- 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE
- 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA) AND ASN FOR A PROPORTION DEFECTIVE (P) BETWEEN AQL AND LTFD
- 6) TO START A NEW PROBLEM
- 7) TO EXIT FROM THE PROGRAM


ENTER YOUR SELECTION OF THE ABOVE OPTIONS ? 3

THE PROGRAM PLOTS FIVE POINTS OF THE OC CURVE. THREE OF THESE CORRESPOND TO THE VALUES OF PROPORTION DEFECTIVE EQUAL TO AQL, SLOPE(S), AND LTFD. THE FOURTH POINT IS IN BETWEEN AQL AND S, AND THE FIFTH IN BETWEEN S AND LTFD.

PRESS ANY KEY TO CONTINUE


OPERATING CHARACTERISTIC CURVE

Y-AXIS: PROBABILITY OF ACCEPTANCE
X-AXIS: FRACTION DEFECTIVE



THE AVERAGE SAMPLE NUMBER CURVE

Y-AXIS : AVERAGE SAMPLE NUMBER
 X-AXIS : FRACTION DEFECTIVE



PRESS ANY KEY TO CONTINUE

AVAILABLE OPTIONS

- 1) TO PRINT A GRAPHICAL REPRESENTATION OF THE PLAN
PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF
YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS.
 - 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE
AND REJECTION NUMBERS
 - 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE
 - 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE
 - 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA)
AND ASN FOR A PROPORTION DEFECTIVE (P)
BETWEEN AQL AND LTFD
 - 6) TO START A NEW PROBLEM
 - 7) TO EXIT FROM THE PROGRAM
-

ENTER YOUR SELECTION OF THE ABOVE OPTIONS ? 5

THIS OPTION TAKES A LITTLE TIME. PLEASE HAVE PATIENCE.

ENTER PROPORTION DEFECTIVE (P).....? .03
FOR PROPORTION DEFECTIVE= 3.099396E-02 PROB. OF ACC.= .6765621 ASN= 74.1681,

PRESS ANY KEY TO CONTINUE

AVAILABLE OPTIONS

- 1) TO PRINT A GRAPHICAL REPRESENTATION OF THE PLAN
PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF
YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS.
- 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE
AND REJECTION NUMBERS
- 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE
- 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE
- 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA)
AND ASN FOR A PROPORTION DEFECTIVE (P)
BETWEEN AQL AND LTFD
- 6) TO START A NEW PROBLEM
- 7) TO EXIT FROM THE PROGRAM

ENTER YOUR SELECTION OF THE ABOVE OPTIONS ? 7

THANKS FOR USING THIS PROGRAM.

5. Conclusion

This paper has presented an interactive computer program in BASIC language for any IBM-PC compatible computer for solving any problem related to an item-by-item sequential sampling plan of the attribute type. The program includes the following regular features

- a) Drawing the plan diagram.
- b) Preparing table of acceptance and rejection numbers.

- c) Drawing the *OC* curve.
- d) Drawing the *ASN* curve (seven points).

In addition, it has a unique feature of calculating the probability of acceptance and the average sampling number for any value of proportion defective between the acceptable quality level and the lot tolerance fraction defective.

References

- [1] **Duncan, A.J.**, *Quality Control and Industrial Statistics*, Homewood, Illinois, Richard D. Irwin Inc. (1974)
- [2] **Montgomery, D.C.**, *Introduction to Statistical Quality Control*, New York, John Wiley and Sons (1985).
- [3] **Rourke, R.S.**, Operating characteristics curves determine acceptance /rejections risks, *Industrial Engineering*, **18** (8) Aug.: 22-24 (1986).
- [4] **Shirtland, L.E.**, Acceptance sampling plan design with risks for consumers and producers, *Industrial Engineering*, **19** (6) June: 27-33 (1987).
- [5] Collins, L., Spoerer, J.P. and **Banks, J.**, *IBM PC Applications for the Industrial Engineer and Manager*. Englewood Cliffs, N.J., Prentice-Hall (1986).
- [6] **Burney, F.A.** and **Abdul Haq, A.K.M.**, Program for double sampling plan by attributes, *Industrial Engineering* (unpublished).
- [7] **Wald, A.**, *Sequential Analysis*, New York, Johny Wiley and Sons (1947).

Program Listing

```

10 PRINT TAB(5)"*****"
20 PRINT TAB(5)"*          PROGRAM FOR SEQUENTIAL SAMPLING PLAN          *"
30 PRINT TAB(5)"*          DEVELOPED BY          *"
40 PRINT TAB(5)"*          DR. F.A.BURNEY          *"
50 PRINT TAB(5)"*          ASSOCIATE PROFESSOR          *"
60 PRINT TAB(5)"*          AND          *"
70 PRINT TAB(5)"*          DR. A.K.M.ABDUL HAQ          *"
80 PRINT TAB(5)"*          ASSISTANT PROFESSOR          *"
90 PRINT TAB(5)"*          INDUSTRIAL ENGINEERING DEPARTMENT          *"
100 PRINT TAB(5)"*          COLLEGE OF ENGINEERING          *"
110 PRINT TAB(5)"*          KING ABDULAZIZ UNIVERSITY          *"
120 PRINT TAB(5)"*          JEDDAH, SAUDI ARABIA          *"
130 PRINT TAB(5)"*****"
140 PRINT : PRINT
150 DIM Z$(20) , P$(200) , XA(50) , YA(50) , X(51) , Y(50) , PR$(51)
160 PRINT "NEED ANY INTRODUCTION (Y OR N) .... ":
170 INPUT T$
180 IF T$ <> "Y" THEN 370
190 PRINT : PRINT
200 REM   *** INTRODUCTION TO THE PLAN   ***
210 PRINT TAB(10)"THIS PROGRAM IS ABOUT SEQUENTIAL SAMPLING PLAN"
220 PRINT TAB(10)"AND IS BASED ON WALD'S SEQUENTIAL PROBABILITY RATIO."
230 PRINT TAB(10)"GIVEN THE VALUES OF THE PRODUCERS RISK(ALPHA)."
240 PRINT TAB(10)"THE CONSUMERS RISK(BETA), THE ACCEPTABLE QUALITY"
250 PRINT TAB(10)"LEVEL,AND THE LOT TOLERANCE FRACTION DEFECTIVE."
260 PRINT TAB(10)"THE PLAN WILL BE FULLY DEFINED BY THE EQUATIONS"
270 PRINT TAB(10)"OF THE FOLLOWING TWO LINES."
280 PRINT :PRINT TAB(10)" LINE OF ACCEPTANCE "
290 PRINT :PRINT TAB(10)" LINE OF REJECTION "
300 PRINT : PRINT
310 PRINT TAB(10)"INPUT REQUIREMENTS"
320 PRINT TAB(10)"-----"
330 PRINT TAB(10)" 1.- ACCEPTABLE QUALITY LEVEL (AQL)"
340 PRINT TAB(10)" 2.- PRODUCER'S RISK (ALPHA VALUE)"
350 PRINT TAB(10)" 3.- LOT TOLERANCE FRACTION DEFECTIVE (LTFD)"
360 PRINT TAB(10)" 4.- CONSUMER'S RISK (BETA VALUE)"
370 REM   I N P U T   S E C T I O N
380 PRINT : PRINT
390 PRINT "ENTER ACCEPTABLE QUALITY LEVEL (AQL)....":
400 INPUT P1
410 IF P1 >=0 AND P1 <=1 THEN 450
420 GOSUB 630
430 PRINT :PRINT "*** A.Q.L MUST BE >=0 & <=1 *** "
440 GOTO 390
450 PRINT "ENTER PRODUCER'S RISK (ALPHA VALUE)....":
460 INPUT AA
470 IF AA >=0 AND AA <=1 THEN 510
480 GOSUB 630
490 PRINT :PRINT " *** ALPHA VALUE MUST BE >=0 & <=1 *** "
500 GOTO 450
510 PRINT "ENTER LOT TOLERANCE FRACTION DEFECTIVE (LTFD)...":
520 INPUT P2
530 IF P2 > P1 AND P2 <=1 THEN 570
540 GOSUB 630
550 PRINT :PRINT " *** L T F D MUST BE > A Q L >=0 & <= 1 *** "
560 GOTO 510

```

```

570 PRINT "ENTER CONSUMER'S RISK (BETA VALUE).....":
580 INPUT BB
590 IF BB >=0 AND BB <=1 THEN 660
600 GOSUB 630
610 PRINT :PRINT " *** BETA VALUE MUST BE >= 0 & <= 1 *** "
620 GOTO 570
630 PRINT " *** E R R O R *** PLEASE TRY AGAIN *** "
640 RETURN
650 REM C A L C U L A T I O N S E C T I O N
660 A = LOG ((1-AA) /BB)
670 B = LOG ((1-BB) /AA)
680 G1 = LOG (P2/P1)
690 G2 = LOG ((1-P1)/(1-P2))
700 H1 = A/ (G1+G2) : REM H1 = THE INTERCEPT OF ACCEPTANCE LINE
710 H2 = B/ (G1+G2) : REM H2 = THE INTERCEPT OF REJECTION LINE
720 S = G2/ (G1+G2) : REM S = THE SLOPE OF THESE LINES
730 H = H2/ (H1 + H2) : REM H = THE PROB. OF ACCEP. FOR LTFD=S
740 REM SELECT OPTION HERE
750 PRINT :PRINT TAB(10)" AVAILABLE OPTIONS "
760 PRINT TAB(10)"-----"
770 PRINT :PRINT " 1) TO PRINT A GRAPHICAL REPRESENTATION OF TH PLAN"
780 PRINT " PLEASE NOTE THAT THIS OPTION MAY BE USED ONLY IF "
790 PRINT " YOUR COMPUTER IS EQUIPPED WITH COLOR GRAPHICS."
800 PRINT :PRINT " 2) TO PRINT AN ILLUSTRATION TABLE OF ACCEPTANCE "
810 PRINT " AND REJECTION NUMBERS"
820 PRINT :PRINT " 3) TO PLOT THE OPERATING CHARACTERISTIC CURVE "
830 PRINT :PRINT " 4) TO PLOT THE AVERAGE SAMPLE NUMBER CURVE "
840 PRINT :PRINT " 5) TO FIND THE PROBABILITY OF ACCEPTANCE (PA)"
850 PRINT :PRINT " AND ASN FOR A PROPORTION DEFECTIVE (P)"
860 PRINT :PRINT " BETWEEN AQL AND LTFD"
870 PRINT :PRINT " 6) TO START A NEW PROBLEM "
880 PRINT :PRINT " 7) TO EXIT FROM THE PROGRAM "
890 PRINT :PRINT
900 PRINT "ENTER YOUR SELECTION OF THE ABOVE OPTIONS ":
910 INPUT SELECTION
920 IF SELECTION = 1 THEN 1030
930 IF SELECTION = 2 THEN 1400
940 IF SELECTION = 3 THEN 1670
950 IF SELECTION = 4 THEN 2290
960 IF SELECTION = 5 THEN 3230
970 IF SELECTION = 6 THEN 370
980 IF SELECTION = 7 THEN 3500
990 IF SELECTION < 1 OR SELECTION > 6 THEN 1000
1000 PRINT " I N V A L I D S E L E C T I O N ! PLEASE TRY AGAI."
1010 PRINT :PRINT " *** S E L E C T I O N IS FROM 1 TO 6 ***"
1020 PRINT :GOTO 750
1030 REM PRINTING GRAPH OF THE SEQ.SAMPLING PLAN
1040 PRINT :PRINT :PRINT
1050 PRINT "IN THE DIAGRAM BELOW THERE ARE THREE REGIONS NAMELY:"
1060 PRINT "ACCEPTANCE REGION BELOW THE BOTTOM LINE, REJECTION REGION"
1070 PRINT "ABOVE THE TOP LINE, AND CONTINUE-SAMPLING REGION IN"
1080 PRINT "BETWEEN THE TWO LINES. FURTHERMORE, THE X-AXIS INDICATES"
1090 PRINT "THE NUMBER OF ITEMS SAMPLED AND THE Y-AXIS THE NUMBER"
1100 PRINT "OF DEFECTIVES."
1110 PRINT "PLEASE ALSO NOTE THAT THIS DIAGRAM CAN BE PRINTED BY"

```

```

1120 PRINT "USING THE CAPSHIFT AND PRISC KEYS TOGETHER."
1130 GOSUB 3190
1140 SCREEN 1 :CLS
1150 KEY OFF
1160 LINE (30,1)-(30,180)
1170 LINE (30,180)-(320,180)
1180 FOR I = 30 TO 280 STEP 40
1190 LINE (I,180)-(I,185)
1200 NEXT I
1210 FOR J = 1 TO 180 STEP 32
1220 LINE (30,J+20)-(25,J+20)
1230 NEXT J
1240 LINE (30,INT (180-36*H2))-(320, INT (180-36*(H2+70*S)))
1250 LINE (INT((290*H1)/(70*S)+30),180)-(320,INT(180-36*(-H1+70*S)))
1260 FOR R =1-TO 17 STEP 4
1270 LOCATE 24-R,1 :PRINT INT ((R-1)/4)
1280 Y$ = Y$ + " "
1290 NEXT R
1300 FOR J = 1 TO 70 STEP 10
1310 LOCATE 24,J/2+2 :PRINT J-1;
1320 NEXT J
1330 LOCATE 8,10 :PRINT "REJ. REGION"
1340 LOCATE 22,30 :PRINT "ACC. REGION"
1350 GOSUB 3190
1360 CLS
1370 WIDTH 80
1380 SCREEN 0
1390 GOTO 750
1400 REM PRINTING THE TABLE OF ACCEPTANCE & REJECTION NOS.
1410 PRINT "-----"
1420 PRINT "TABLE OF ACCEPTANCE & REJECTION NOS. FOR SEQUENTIAL SAMPLING"
1430 PRINT "PLAN WITH AQL=";P1",LTFD=";P2",ALPHA=";AA",BETA=";BB;"AND "
1440 PRINT "UPTO A SAMPLE SIZE OF 80.IN THE TABLE BELOW N IS THE SAMPLE "
1450 PRINT "SIZE, A THE ACCEPTANCE NUMBER, AND R THE REJECTION NUMBER."
1460 PRINT "-----"
1470 PRINT " : N A R : N A R : N A R : N A R "
1480 PRINT "-----"
1490 N1=1
1500 FOR N2=1 TO 4
1510 FOR N3=1 TO 5
1520 FOR NN =1 TO 4
1530 D1 = INT (S * N1 - H1) : REM D1 = ACCEPTANCE NUMBER
1540 D2 = INT (S * N1 + H2+.9990001) :REM D2 = REJECTION NUMBER
1550 IF D1 < 0 GOTO 1560 ELSE GOTO 1580
1560 PRINT USING " : ## * # " :N1, D2;
1570 GOTO 1590
1580 PRINT USING " : ## ## # " :N1,D1,D2;
1590 N1=N1+1
1600 NEXT NN
1610 PRINT
1620 NEXT N3
1630 PRINT "-----"
1640 NEXT N2
1650 GOSUB 3190
1660 GOTO 750

```

```

1670 REM CALCULATION OF FIVE POINTS OF OPERATING CHARACTERISTIC CURVE
1680 X(1) = P1 : Y(1) = 1.- AA
1690 X(3) = S : Y(3) = H
1700 X(5) = P2 : Y(5) = BB
1710 I = 6
1720 FOR W = 1 TO 2
1730 WW = W -1.5
1740 I = I - 2
1750 PPP = (( 1-P2) / (1-P1))^WW
1760 PFA = ((1-BB) / AA)^WW
1770 APP = (BB/ (1-AA))^WW
1780 PP = (P2/P1)^WW
1790 X(I) = (1-PPP) / (PP-PPP)
1800 Y(I) = (PFA-1) / (PFA-APP)
1810 PRINT
1820 NEXT W
1830 REM PLOTTING THE O.C. CURVE
1840 PRINT :PRINT :PRINT
1850 PRINT "THE PROGRAM PLOTS FIVE POINTS OF THE OC CURVE. THREE OF "
1860 PRINT "THESE CORRESPOND TO THE VALUES OF PROPORTION DEFECTIVE "
1870 PRINT "EQUAL TO AQL, SLOPE(S), AND LTFD. THE FOURTH POINT IS IN "
1880 PRINT "BETWEEN AQL AND S. AND THE FIFTH IN BETWEEN S AND LTFD."
1890 GOSUB 3190
1900 F1$ = "#.### "
1910 F2$ = " ###.### "
1920 PRINT : PRINT
1930 PRINT TAB(25); "OPERATING CHARACTERISTIC CURVE "
1940 PRINT : PRINT
1950 PRINT "Y-AXIS: PROBABILITY OF ACCEPTANCE"
1960 PRINT "X-AXIS: FRACTION DEFECTIVE "
1970 PRINT
1980 PRINT TAB(37); " Y - A X I S "
1990 PRINT
2000 PRINT TAB(22);
2010 FOR I = 0 TO 1 STEP .2
2020 PRINT USING F1$:I;
2030 NEXT I
2040 PRINT
2050 PRINT " X VALUES +":
2060 FOR I = 1 TO 5
2070 PRINT "-----+":
2080 NEXT I
2090 PRINT
2100 PR$(0) = "I"
2110 FOR I = 1 TO 5
2120 FOR JK = 1 TO 50
2130 PR$(JK) = " "
2140 NEXT JK
2150 KP=INT(50*(Y(I)) +.99999)
2160 PR$(KP) = "*"
2170 PRINT USING F2$:X(I);
2180 FOR JKK = 0 TO KP
2190 PRINT PR$(JKK);
2200 NEXT JKK
2210 KT = INT (200*(X(I+1)-X(I)))
2220 FOR PNI = 1 TO KT

```



```

2230 PRINT
2240 NEXT PNI
2250 NEXT I
2260 PRINT : PRINT
2270 GOSUB 3190
2280 GOTO 750
2290 REM CALCULATING SEVEN POINTS FOR ASN CURVE
2300 PRINT :PRINT" THIS OPTION TAKES A LITTLE TIME. PLEASE HAVE PATIENCE."
2310 PRINT :PRINT
2320 I = 7
2330 FOR W = 1 TO 2
2340 WW = W-1.5
2350 I = I-2
2360 PPP = ((1-P2)/(1-P1))^WW
2370 PPA = ((1-BB)/AA)^WW
2380 APP = (BB/(1-AA))^WW
2390 PP = (P2/P1)^WW
2400 X(I) = (1-PPP)/(PP-PPP)
2410 Y(I) = (PPA-1)/(PPA-APP)
2420 C = LOG(BB/(1-AA))
2430 D = LOG((1-P2)/(1-P1))
2440 YA(I) = (Y(I)*C+(1-Y(I))*B)/(X(I)*G1+(1-X(I))*D)
2450 XA(I) = X(I)
2460 NEXT W
2470 XA(1) = 0 : YA(1) = H1/S
2480 XA(2) = P1 : YA(2) = (H1*(1-AA)- AA*H2) / (S-P1)
2490 XA(4) = S : YA(4) = H1*H2 / (S* (1-S))
2500 XA(6) = P2 : YA(6) = (H2* (1-BB) - BB*H1) ./ (P2-S)
2510 PL = P1+P2
2520 FOR W = 500 TO 1000
2530 WW = (W/500)-3
2540 PPP = ((1-P2)/(1-P1))^WW
2550 PP = (P2/P1)^WW
2560 IF PP = PPP GOTO 2590
2570 PD = (1-PPP)/(PP-PPP)
2580 IF ABS(PD-PL) <= .001 THEN 2600
2590 NEXT W
2600 PPA = ((1-BB)/AA)^WW
2610 APP = (BB/(1-AA))^WW
2620 PA = (PPA-1)/(PPA-APP)
2630 C = LOG(BB/(1-AA))
2640 D = LOG((1-P2)/(1-P1))
2650 ASN = (PA*C+(1-PA)*B)/(PD*G1+(1-PD)*D)
2660 XA(7) = P1+P2 : YA(7) = ASN
2670 Z=YA(1)
2680 FOR I = 2 TO 7
2690 IF Z < YA(I) THEN Z = YA(I)
2700 NEXT I
2710 REM PLOTTING THE ASN CURVE
2720 PRINT :PRINT :PRINT
2730 PRINT "THE PROGRAM PLOTS SEVEN POINTS OF THE ASN CURVE. THESE "
2740 PRINT "CORRESPOND TO THE VALUES OF PROPORTION DEFECTIVE EQUAL "
2750 PRINT "TO ZERO. AQL. BETWEEN AQL AND S, S, BETWEEN S AND LTFD."
2760 PRINT "LTFD. AND P1+P2."
2770 GOSUB 3190

```

```

2780 Z$(1) = "### "
2790 Z$(2) = " ###.### "
2800 PRINT :PRINT
2810 PRINT TAB(10):" THE AVERAGE SAMPLE NUMBER CURVE"
2820 PRINT :PRINT
2830 PRINT "Y-AXIS : ""AVERAGE SAMPLE NUMBER"
2840 PRINT "X-AXIS : ""FRACTION DEFECTIVE "
2850 PRINT
2860 PRINT TAB(37):"Y - A X I S      "
2870 PRINT      - - -
2880 PRINT TAB(10):
2890 AD = 5*(INT(Z/50))
2900 DR = 5+ AD
2910 FOR I = 0 TO Z+10 STEP DR
2920 PRINT USING Z$(1);I;
2930 NEXT I
2940 PRINT
2950 PRINT "X-A X I S +";
2960 FOR I = 1 TO (Z+10) / DR
2970 PRINT "-----+":
2980 NEXT I
2990 PRINT
3000 P$(0) = ":"
3010 FOR I = 1 TO 7
3020 FOR JK = 1 TO 60
3030 P$(JK) = " "
3040 NEXT JK
3050 FG = INT ((Z/50)+.99999)
3060 PT = INT ((YA(I) /FG) +.99999)
3070 P$(PT) = "*"
3080 PRINT USING Z$(2): XA(I);
3090 FOR JKK = 0 TO PT
3100 PRINT P$(JKK):
3110 NEXT JKK
3120 PK = INT(200*(XA(I+1)-XA(I)))
3130 FOR PKT = 1 TO PK
3140 PRINT
3150 NEXT PKT
3160 NEXT I
3170 GOSUB 3190
3180 GOTO 750
3190 PRINT : PRINT
3200 PRINT TAB(20) "PRESS ANY KEY TO CONTINUE"
3210 Z$ = INKEY$ : IF Z$="" THEN 3210
3220 RETURN
3230 REM CALCULATION OF PA AND ASN GIVEN PROPORTION DEFECTIVE
3240 PRINT :PRINT" THIS OPTION TAKES A LITTLE TIME. PLEASE HAVE PATIENCE."
3250 PRINT :PRINT
3260 PRINT "ENTER PROPORTION DEFECTIVE (P).....":
3270 INPUT P
3280 IF P>=P1 AND P<=P2 THEN 3320
3290 GOSUB 630
3300 PRINT :PRINT"*** P MUST BE >=AQL & <=LTFD *** "
3310 GOTO 3260
3320 FOR W = 500 TO 1500
3330 WW = (W/500)-2

```

```
3340 PPP = ((1-P2)/(1-P1))^WW
3350 PP = (P2/P1)^WW
3360 IF PP = PPP GOTO 3390
3370 PD = (1-PPP)/(PP-PPP)
3380 IF ABS(PD-P) <= .001 THEN 3400
3390 NEXT W
3400 PPA = ((1-BB)/AA)^WW
3410 APP = (BB/(1-AA))^WW
3420 PA = (PPA-1)/(PPA-APP)
3430 C = LOG(BB/(1-AA))
3440 D = LOG((1-P2)/(1-P1))
3450 ASN = (PA*C+(1-PA)*D)/(PD*D+(1-PD)*D)
3460 PRINT "FOR PROPORTION DEFECTIVE=":PD;"PROB. OF ACC.=":PA;"ASN ":ASN
3470 PRINT :PRINT
3480 GOSUB 3190
3490 GOTO 750
3500 PRINT
3510 PRINT "THANKS FOR USING THIS PROGRAM."
```

برنامج لسحب العينات تتابعياً بالصفات المميزة لكل عينة

فرحات علي برني و أبو الكلام محمد عبد الحق
 قسم الهندسة الصناعية ، كلية الهندسة ، جامعة الملك عبد العزيز
 جدة - المملكة العربية السعودية .

المستخلص . يقوم البرنامج المصمّم بلغة البيسك ليناسب الحاسبات الشخصية بحل مسائل سحب العينات تتابعياً ، وذلك إذا أعطيت مستوى الجودة المسموح به ودرجة المخاطرة لكل من المستهلك والمنتج وكذا سماح المعيوبية .

ويقوم هذا البرنامج بإنتاج المخرجات الآتية :

- ١ - عرض الخطة بالأشكال والرسوم الملونة (لوسمح الجهاز المستخدم بذلك) .
- ٢ - جداول إعداد القبول والرفض .
- ٣ - منحي الخصائص التشغيلية .
- ٤ - منحي متوسط رقم العينة .

كما ويحظي هذا البرنامج بصفة فريدة حيث أنه يقوم بحساب احتمال قبول العينات وكذا متوسط رقم العينة لأي قيمة لدرجة المعيوبية الشبيهة بين مستوى الجودة المقبول والعيوب المسموح بها .